
MATHEMATICS OF COMPUTATION
Volume 79, Number 270, April 2010, Pages 1091–1108
S 0025-5718(09)02283-2
Article electronically published on September 2, 2009

ON EQUATIONS OF DOUBLE PLANES WITH pg = q = 1

CARLOS RITO

Abstract. This paper describes how to compute equations of plane models
of minimal Du Val double planes of general type with pg = q = 1 and K2 =
2, . . . , 8. A double plane with K2 = 8 having bicanonical map not composed
with the associated involution is also constructed. The computations are done
using the algebra system Magma.

1. Introduction

There are two typical methods to construct examples of surfaces: Campedelli
— double covers ramified over curves, usually highly singular — and Godeaux —
quotients by group actions (cf. [Rei91]). In this paper the first method is used
to obtain new examples of double planes of general type with pg = q = 1. Often
the singularities impose too many conditions on the linear parameters of systems
of curves, and this implies hard calculations. The Computational Algebra System
MAGMA ([BCP97]) is used to perform the computations.

Several authors have studied surfaces of general type with pg = q = 1 ([Cat81],
[Cat99], [CC91], [CC93], [CP06], [Pol08], [Pol09], [Pol05], [Pol06], [Pig]), but these
surfaces are still not completely understood, and few examples are known.

For such a smooth minimal surface S, one has 2 ≤ K2 ≤ 9 and K2 = 9 only if
the bicanonical map of S (given by |2K|) is birational (from [CM02] and [Xia85,
Théorème 2.2]). Since q = 1, the Albanese map of S is a connected fibration onto
an elliptic curve. We denote by g the genus of a general Albanese fibre of S.

Du Val ([Du 52]) classified the regular surfaces S of general type with pg ≥ 3
whose general canonical curve is smooth and hyperelliptic. For these surfaces the
bicanonical map is composed with an involution i such that S/i is rational. The
families of surfaces exhibited by Du Val are nowadays called the Du Val examples.

The standard case of non-birationality of the bicanonical morphism φ2 is the
case where S has a genus 2 fibration.

For the non-standard case with φ2 of degree 2, bicanonical image a ruled surface
and pg(S) ≥ 1, Gang Xiao ([Xia90, Theorem 2]) presented a list of possibilities
which is a generalization of Du Val’s list plus two extra families (this result is still
valid assuming only that φ2 is composed with an involution such that the quotient
surface is a ruled surface).

Later G. Borrelli ([Bor07]) excluded these two families and confirmed that, if
the bicanonical map φ2 of S is composed with i, S/i is ruled and S presents the
non-standard case, then S is a Du Val double plane (see Definition 3). But, to my
knowledge, the existence of these double planes with pg = q = 1 and K2 = 4, . . . , 7
has not yet been shown, and no equation has been given for K2 = 3 or 8.

In this paper we compute equations of plane models of minimal Du Val double
planes of general type with pg = q = 1 and K2 = 2, . . . , 8. We also construct a
double plane withK2 = 8 having bicanonical map not composed with the associated
involution.

Notice that we give the first example of a minimal surface of general type with
pg = q = 1 and K2 = 7. We also give the first example with K2 = 5 having
Albanese fibration of genus g �= 2.

The first example with K2 = 6 has been given by the author in [Rit07], as a
double cover of a K3 surface. Polizzi ([Pol09]) also gives examples with K2 = 6, 4
and g �= 2. These surfaces contain 8 − K2 (−2)-curves, all in the same Albanese
fibre. One can verify that, in the double plane examples with K2 = 4 or 6, there
is no Albanese fibre containing 8 − K2 (−2)-curves. Therefore our surfaces are
different from the above examples.

In [Pol06], Polizzi classifies surfaces of general type with pg = q = 1, K2 = 8
and bicanonical map of degree 2. He constructs examples using quotients under the
action of a group and shows that these surfaces are Du Val double planes, describing
the branch locus of the respective plane models. In sections 3.1 and 3.4 we show
how to obtain equations for such branch loci.

Section 2 contains a structure description of the Albanese fibration of S and the
definition of a Du Val double plane. In Section 3 we describe the principal steps of
the constructions. To obtain the equations of the branch loci we impose conditions
to the linear parameters of systems of plane curves. Since the ramification curves
are contained in Albanese fibres, in some cases it is easier to start by constructing
non-reduced Albanese fibres, which simplify the computations. These are done
using the algebra system Magma in Appendix A.

Notation and conventions. We work over the complex numbers; all varieties are
assumed to be projective algebraic. For a projective smooth surface S, the canonical
class is denoted by K, the geometric genus by pg := h0(S,OS(K)), the irregularity
by q := h1(S,OS(K)) and the Euler characteristic by χ = χ(OS) = 1 + pg − q.

A (−n)-curve C on a surface is a curve isomorphic to P
1 such that C2 = −n. We

say that a curve singularity is negligible if it is either a double point or a triple point
which resolves to at most a double point after one blowup. An (m1,m2, . . .)-point,
or point of order (m1,m2, . . .), is a point of multiplicity m1, which resolves to a
point of multiplicity m2 after one blowup, etc.

An involution of a surface S is an automorphism of S of order 2. We say that
a map is composed with an involution i of S if it factors through the double cover
S → S/i.

The rest of the notation is standard in algebraic geometry.

2. Preliminaries

We say that a smooth surface S is a double plane if S has an involution i such
that S/i is a rational surface. A plane model of S is a double cover X → P

2 such

that X is a normal surface and there exists a commutative diagram

S −−−−→ X⏐⏐�
⏐⏐�

S/i −−−−→ P
2

such that the horizontal arrows denote birational maps.

2.1. Double planes with q = 1. Let S be a smooth minimal surface of general
type with an involution i. Since S is minimal of general type, this involution is
biregular. The fixed locus of i is the union of a smooth curve R′′ (possibly empty)
and of t ≥ 0 isolated points P1, . . . , Pt. Let S/i be the quotient of S by i and
p : S → S/i be the projection onto the quotient. The surface S/i has nodes at the
points Qi := p(Pi), i = 1, . . . , t, and is smooth elsewhere. If R′′ �= ∅, the image via p
of R′′ is a smooth curve B′′ not containing the singular points Qi, i = 1, . . . , t. Now
let h : V → S be the blowup of S at P1, . . . , Pt and set R′ = h∗(R′′). The involution

i induces a biregular involution ĩ on V, whose fixed locus is R := R′ +
∑t

1 h
−1(Pi),

and the quotient W := V/̃i is smooth.
Let B be the branch locus of the projection π : V → W, i.e. the image of R on

W. Let L be the line bundle such that 2L ≡ B and B,L determine π. We have

(1) H0(V,OV (2KV )) ∼= H0(W,OW (2KW +B))⊕H0(W,OW (2KW + L)).

The first and second summands correspond, respectively, to the invariant and anti-
invariant parts of the bicanonical system of V.

Suppose now that q(S) = 1 and that S/i is a rational surface. Since q = 1, the
Albanese variety of S is an elliptic curve E and the Albanese map is a connected
fibration (see e.g. [Bea78] or [BPV84]). This fibration is preserved by i and so we
have a commutative diagram

(2)

V
h−−−−→ S −−−−→ E

π

⏐⏐� p

⏐⏐�
⏐⏐�

W −−−−→ S/i −−−−→ P
1.

Denote by

fA : W → P
1

the fibration induced by the Albanese fibration of S. The double cover E → P
1 is

ramified over 4 points pj of P
1; thus the branch locus B of π is contained in 4 fibres

F j
A := f∗

A(pj), j = 1, ..., 4,

of the fibration fA. By Zariski’s Lemma (see e.g. [BPV84]) the irreducible compo-

nents Bi of B satisfy B2
i ≤ 0. Since π∗(F j

A) has even multiplicity, each component

of F j
A which is not a component of the branch locus must be of even multiplicity.

In the following sections W can always be contracted to P
2. We keep the same

notation for the image of fA and F j
A on P

2.

2.2. Du Val double planes.

Notation 1. Given distinct points p0, . . . , pj , . . . , pj+s ∈ P
2, let Ti denote the line

through p0 and pi, i = 1, . . . , j. We say that a plane curve is of type

d
(
m, (n, n)jT , r

s
)

if it is of degree d and if it has an m-tuple point at p0, an (n, n)-point at p1, . . . , pj ,
an r-tuple point at pj+1, . . . , pj+s and no other non-negligible singularities. The
index T is used if Ti is tangent to the (n, n)-point at pi.

An obvious generalization is used if there are other singularities.

Definition 2. Let C0 and F denote, respectively, the negative section and a rational
fibre of the Hirzebruch surface F2.

The Bombieri-Du Val surface is the minimal model of the double cover of F2

with branch locus a smooth curve in C0 + |7C0 + 14F |.

Definition 3. A Du Val surface is either

B) the Bombieri-Du Val surface

or a double plane having a plane model X → P
2 with branch locus D one of the

following:

D) a smooth curve of degree 8;
D0) a smooth curve of degree 10;
Dn) a curve of type [10 + 2n]

(
2n+ 2, (5, 5)nT

)
, n ∈ {1, . . . , 6};

or one of B, D or Dn, imposing additional singularities to D which correspond, on
the branch locus of the canonical resolution of X, to negligible singularities, 4-tuple
points or (3, 3)-points.

Surfaces of type B, D or Dn are called Du Val’s ancestors. Du Val surfaces are
also called Du Val double planes. We have the following:

Theorem 4 ([Bor07]). Let S be a smooth minimal surface of general type. Suppose
that S has no pencil of genus-2 curves. Then the following conditions are equivalent:

a) the bicanonical map of S factors through a rational map of degree two φ :
S ��� Σ, where Σ is a rational or ruled surface;

b) S is the smooth minimal model of a Du Val double plane.

Moreover, if S is as in b), then q(S) = 0 unless pg(S) = q(S) = 1.

Notice that the imposition of a 4-tuple point to the branch locus decreases K2

by 2 and the Euler characteristic χ by 1, while a (3, 3)-point decreases both K2 and
χ by 1. Negligible singularities in the branch locus do not change these invariants.

The imposition of 6− n 4-tuple or (3, 3)-points to the branch locus of a Du Val
surface of type Dn gives pg = q = 1 if there is exactly one conic through the singular
points p1, . . . , p6.

In the case Dn, the branch locus contains the lines T1, . . . , Tn. Each of these
lines corresponds to two (−2)-curves in W, which contract to two nodes of S/i. A
(3, 3)-point in the branch locus also corresponds to a node of S/i.

The bicanonical map of a Du Val double plane of general type factors through
a map of degree 2 onto a rational surface because the anti-invariant part of the
bicanonical system is empty. Moreover, using (1) of Section 2.1 and using that
h0(S,OS(2KS)) = K2

S + χ(S) for S minimal of general type (see e.g. [BPV84, Ch.
VII, Cor. 5.4]), one can show the following:

Proposition 5. Let S be the minimal model of a Du Val double plane obtained by
imposing (3, 3) or 4-tuple points to the branch locus D of a Du Val’s ancestor of
type Dn. Let a, b be the number of such points, respectively.

If S is a surface of general type, then K2
S + χ(S) is equal to the number of

generators of the linear system∣∣∣∣∣(4 + 2n)ψ∗(H)− 2nE0 −
n∑
1

(2Ei + 4E′
i)−

n+a∑
n+1

2E′
i −

n+a+b∑
n+a+1

2Ei

∣∣∣∣∣ ,

where ψ is the blowup of P2 which resolves the singularities of D, H is a line in P
2,

E0, . . . , En+a+b are exceptional divisors with self-intersection −1 and E′
1, . . . , E

′
n+a

are (−1)-curves corresponding to the blowup at the infinitely near points.

For more information on Du Val surfaces, see [Du 52], [Cil97] or [Bor07].

2.3. The Magma procedures LinSys and LinSys2. Magma has the function
LinearSystem, which calculates linear systems of plane curves with ordinary sin-
gularities, but we want to work with non-ordinary singularities. To achieve this,
we define two procedures. The first one, LinSys, calculates the linear system L of
plane curves of degree d, in an affine plane A, having singular points pi of order
(m1i,m2i) with tangent slope tdi.

The other procedure, LinSys2, calculates the subsystem J, of a given linear
system L of plane curves, of those sections which have a singularity, at a point q,
of type m = (m1, . . . ,mj) with tangent slopes given by td = [td1, . . . , tdj−1].

The code lines for these procedures are in Appendix A.1.

3. Equations of plane models with pg = q = 1

In this section we obtain equations of the branch locus B ⊂ P
2 = P(1, 1, 1) of

minimal Du Val double planes S, with ancestor of type Dn (see Definition 3), with
pg = q = 1 and K2 = 2, . . . , 8. Notice that if f = 0 is the defining equation of B,
then the equation w2 = f gives a plane model of S in the weighted projective space
P((deg f)/2, 1, 1, 1).

In Section 3.5 we construct a double plane with K2 = 8 having bicanonical map
not composed with the associated involution.

In order to obtain q = 1, all the singularities, except the one at p0, are chosen
to be contained in exactly one conic.

Let W, L be as in Section 2.1. In each case we compute

K2 = 2(KW + L)2 + t,

where t is the number of (−1)-curves contained in the ramification divisor. We
get K2 > 0. Since also pg > 0, S is of general type. To exclude the possibility of
hidden (−1)-curves, we use the Magma procedure LinSys defined in Appendix A.1
to compute the dimension of the linear system given in Proposition 5. This returns
the same value for K2, which implies the minimality of the surface.

Table 1 lists the type of each branch curve that we are going to obtain and the
corresponding values of (K2, g), where g denotes the genus of a general Albanese
fibre of S.

We keep Notation 1.
First we find double planes with (K2, g) = (8, 5) and (K2, g) = (7, 5). In the

first case the branch locus contains an element of the pencil fA which induces the

Table 1

Type of branch curve (K2, g)

22
(
14, (5, 5)6T

)
(8, 5), (8, 4), (8, 3)

[10 + 2n]
(
2n+ 2, (5, 5)nT , 4

6−n
)

(6, 4), (4, 3), (2, 2)
n = 5, 4, 3

[10 + 2n]
(
2n+ 2, (5, 5)nT , (3, 3), 4

5−n
)

(7, 5), (5, 4), (3, 3)
n = 5, 4, 3

18
(
10, (5, 5)4T , (3, 3)

2
)

(6, 3)

26
(
14, (7, 7)5T

)
(8, 4)

Albanese fibration. From here constructions with (K2, g) = (6, 4), (5, 4), (4, 3),
(3, 3), (2, 2) will follow easily. A construction with (K2, g) = (6, 3) and branch
locus strictly contained in elements of fA is also given. Then we get surfaces with
(K2, g) = (8, 4) or (8, 3). These three surfaces with K2 = 8 are surfaces of type I, II,
III described by F. Polizzi in [Pol06]. Finally we construct a pencil l of plane curves
and a double plane with branch locus containing an element of l having K2 = 8
and bicanonical map not composed with the corresponding involution.

A difficulty that arises in the computation of singular curves is that often the
computer is not able to finish the calculations. Except for Section 3.2, our method
is to try to figure out the configuration of the non-reduced fibres F i

A of fA, described
in Section 2.1. The problem of finding the support of F i

A deals with curves of lower
degree and with simpler singularities, hence with faster computations.

The next sections of this chapter describe the principal steps. The detailed
calculations are done in Appendix A, using the Computational Algebra System
Magma.

We keep Notation 1.

3.1. K2 = 8, 6, 4, 2 and g = 5, 4, 3, 2. To construct a Du Val’s ancestor S of type
D6 (it has K2 = 8) it suffices to find a curve B′ of type

16
(
8, (4, 4)6T

)
,

singular at p0, . . . , p6, such that the branch locus B := B′ +
∑6

1 Ti is reduced. If
p1, . . . , p6 are contained in exactly one conic, then pg(S) = q(S) = 1.

In this example the pencil fA which induces the Albanese fibration of S has the
following elements:

F 1
A = 2D1 + T3 + · · ·+ T6, F 2

A = 2D2 + T1 + T2, F 3
A = 2D3, F 4

A = D4,

where D1, . . . , D4 are curves of types

6
(
2, (2, 2)2T , (2, 1)

4
T

)
, 7

(
3, (2, 1)2T , (2, 2)

4
T

)
, 8

(
4, (2, 2)6T

)
, 16

(
8, (4, 4)6T

)
,

respectively.
To construct B′, we first find points p0, . . . , p6 such that D1 exists. Then we

use the procedure LinSys (see Section 2.3) to verify the existence of D2 and D3,
also singular at p0, . . . , p6. The curve B′ = D4 is a general element of the pencil
generated by F 1

A, F
2
A, F

3
A.

So let us look for D1. Briefly, the steps are as follows. Let A be an affine plane,
C be a smooth conic not containing the origin p0 of A and p1, . . . , p4 be points in
C. Denote by L the linear system of plane curves of type

6
(
2, (2, 2)2T , (2, 1)

2
T

)
,

with singularities at p0, . . . , p4, respectively. Let F be a general element of L and
p5, p6 be general points of A. We define a scheme Sch by imposing the following
conditions:

• p5, p6 ∈ C ∩ F ;
• p5, p6 are double points of F ;
• the singularities of F at p5, p6 have one branch tangent to T5, T6;
• p5 �= p6 and p5, p6 �∈ {p0, . . . , p4}.

Now we compute the points of Sch with Magma, choosing one of the solutions for
p5, p6, and we use the procedure LinSys to compute B′, a general element of the
pencil of curves of type 16(8, (4, 4)6T ), with singularities at p0, . . . , p6.

Finally we perform some verifications: that B′ is reduced, the singularities are
as expected, the value of K2

S is as claimed, etc.

With this we find a minimal double plane with pg = q = 1 and K2 = 8. The
divisor 2D1 + T3 + · · · + Ti is also a good candidate for one of the singular fibres
F i
A in the case where the branch locus is a curve of type

[10 + 2n]
(
2n+ 2, (5, 5)nT , 4

6−n
)
, n = 5, 4, 3.

In fact, using the points p0, . . . , p6 and the procedure LinSys, one can find branch
loci of those types, obtaining then minimal double planes with pg = q = 1,
K2 = 6, 4, 2 and g = 4, 3, 2, respectively.

The corresponding Magma calculations are in Appendix A.2.1. There we use
symmetry in order to obtain faster computations.

3.2. K2 = 7, 5, 3 and g = 5, 4, 3. Here we impose a (3, 3)-point to the branch locus
of a Du Val’s ancestor of type D5; i.e., we construct a plane curve B′ of type

15(7, (4, 4)5T , (3, 3))

such that B := B′ +
∑5

1 Ti is reduced (see Notation 1). The double cover with
branch locus B is a plane model of a Du Val double plane S with K2 = 7 and
χ = 1.

In this example the (3, 3)-point p6 is infinitely near to the (5, 5)-point p1 of B
(i.e., there is a (5, 5, 3, 3)-point at p1) and the line T1, through p0, p1, is tangent to

the conic C defined by p1, . . . , p5. This implies that q(S) = 1, because C̃ −
∑6

1 Ei

is effective, where the curves Ei are the exceptional divisors corresponding to the

blowups at the points pi, and C̃ is the pullback of C.
In this case the pencil fA has elements

F 1
A = 2D1 + T4 + T5, F 2

A = 2D2 + T2 + T3, F 3
A = 2D3, F 4

A = D4 + T1,

where D1, . . . , D4 are curves of types

7
(
3, (2, 2)3T , (2, 1)

2
T , (2, 2)

)
, 7

(
3, (2, 2)T , (2, 1)

2
T , (2, 2)

2
T , (2, 2)

)
,

8
(
4, (2, 2)5T , (2, 2)

)
, 15(7, (4, 4)5T , (3, 3)),

respectively.
To find B′ = D4 we proceed as follows. In an affine plane A, we fix a smooth

conic C not containing the origin p0 and we choose distinct points p1, . . . , p5 ∈ C
such that T1 is tangent to C at p1. We compute the linear system L of plane curves
of type 15(7, (4, 4)5T ) and we resolve the base point of L at p1, denoting the resulting
linear system by L1. The defining polynomial of B′ is given by (the blowdown of)
a linear combination of elements of L1.

In order to obtain a (3, 3)-point p6 (infinitely near to p1), we impose conditions,
given by the zeros of the derivatives up to order 2, to the elements of L1. Then we
resolve this point and impose another triple (infinitely near) point. With all these
conditions we define a matrix, which is denoted by Mt in Appendix A.2.2. To have
a solution it is necessary that Mt has no maximal rank.

We define a scheme Sch by imposing the following conditions:

• the vanishing of the maximal minors of Mt;
• the points are infinitely near.

Now we compute the points of Sch with Magma, choosing one of the solutions for
p6, and we use the procedures of Section 2.3 to compute B′, with singularities at
p0, . . . , p6.

Finally we perform some verifications. In particular we show that B′ is reduced,
the singularities are as expected, there are no other singularities and the value of
K2

S is as claimed.

To find the pencil which induces the Albanese fibration, one uses again the
procedures LinSys and LinSys2 to compute the pencil of curves of type

16(8, (4, 4)5T , (4, 4)),

through p0, . . . , p6, and to verify that the elements F 1
A, . . . , F

4
A are as claimed.

With this we find a minimal double plane with pg = q = 1, K2 = 7 and g = 5.
Using the above Magma procedures, one can verify that there exist also branch loci
of type

18
(
10, (5, 5)4T , 4, (3, 3)

)
and 16

(
8, (5, 5)3T , 4

2, (3, 3)
)
,

through p0, . . . , p6. These correspond to minimal Du Val double planes with pg =
q = 1 and (K2, g) = (5, 4), (3, 3). The pencils which induce the Albanese fibration
are of type

15
(
7, (4, 4)4T , 4, (4, 4)

)
and 14

(
6, (4, 4)3T , 4

2, (4, 4)
)
,

respectively.
The Magma calculations for this section are in Appendix A.2.2.

3.3. K2 = 6 and g = 3. In this section we impose two (3, 3)-points to the branch
locus of a Du Val’s ancestor of type D4; i.e., we construct a plane curve B′ of type

14(6, (4, 4)4T , (3, 3)
2),

singular at points p0, . . . , p6, such that B := B′+
∑4

1 Ti is reduced. In this example
the (3, 3)-points p5, p6 are tangent to the conic C through p1, . . . , p6. The double
cover with branch locus B is a plane model of a Du Val double plane S with K2 = 6,
pg = q = 1 and g = 3.

The pencil fA which induces the Albanese fibration of S contains four fibres

F 1
A = D1 + 2T0, F 2

A = D2, F 3
A = 2D3, F 4

A = 2C + T1 + · · ·+ T4,

where T0 is a line through p0, p5, p6, and D1, D2, D3 are curves of types

6
(
2, (2, 2)4T , (1, 1)

2
)
, 8

(
4, (2, 2)4T , (2, 2)

2
)
, 4

(
2, (1, 1)4T , (1, 1)

2
)
,

respectively. We have
B′ = D1 +D2.

First we construct D1 and then we use the procedure LinSys of Section 2.3 to
obtain D2 (a general element of fA) and to verify the existence of D3.

To find D1, we follow the steps used in Section 3.1, but now with L the linear
system of curves of type 6

(
2, (2, 2)4T

)
and Sch defined by

• p5, p6 ∈ C ∩ F ;
• F is smooth at p5, p6;
• F is tangent to C at p5, p6;
• p5 �= p6 and p5, p6 �∈ {p0, . . . , p4}.

The details can be found in Appendix A.2.3. Again we use symmetry in order
to increase the speed of the calculations.

3.4. K2 = 8 and g = 4 or 3. Given, in an affine plane A, a general plane cubic Q
not containing the origin p0, there are 6 points p1, . . . , p6 ∈ Q such that each line
Ti, defined by p0, pi, is tangent to Q. In this example we have

F 1
A = 3D1, F 2

A = D2, F 3
A = 2D3, F 4

A = 2Q+ T1 + · · ·+ T6,

where D1, D2, D3 are curves of types

4
(
2, (1, 1)6T

)
, 12

(
6, (3, 3)6T

)
, 6

(
3, (2, 1)6T

)
,

respectively.
Using the procedure LinSys (see Section 2.3), one can verify the existence of the

curves D1, D2, D3, through p0, . . . , p6, and that p1, . . . , p6 are contained in exactly
one conic. The double cover with branch locus D1 +D2 +

∑6
1 Ti is a plane model

of a Du Val double plane S with K2 = 6, pg = q = 1 and g = 4. This is a surface
of type II described in [Pol06].

Finally the surface of type I described in [Pol06]:
Let C ⊂ P

2 be a smooth conic and p0 �∈ C, p1, . . . , p6 ∈ C be points such that
the lines T5 and T6, through p0, p5 and p0, p6, are tangent to C. One can verify the
existance of a smooth cubic D of type 3(1, (1, 1)4T , 1

2), through p0, . . . , p6. Let

F 1
A = 2C + T1 + · · ·+ T4, F 2

A = 2D + T5 + T6

and F 3
A, F

4
A be general elements of the pencil generated by F 1

A and F 2
A. The branch

locus in this case is

F 3
A + F 4

A +

6∑
1

Ti.

3.5. K2 = 8, g = 4, φ2 not composed with i. This section describes the steps
taken to obtain an equation of a double plane S of general type with K2 = 8,
pg = q = 1 and Albanese fibres of genus g = 4, such that the bicanonical map φ2

of S is not composed with the associated involution i of S.
In this example the pencil fA which induces the Albanese fibration has four fibres

F 1
A = 2D1 + T5, F 2

A = 6D2 + 3T1, F 3
A = 2D3 + T2 + T3 + T4, F 4

A = D4,

where D1, . . . , D4 are curves of types

10
(
4, (3, 3)4T , (3, 2)T

)
, 3

(
1, 1, (1, 1)4T

)
,

9
(
3, (3, 3), (3, 2)3T , (3, 3)

)
, 21

(
9, (6, 6)5T

)
,

respectively. The surface S is the minimal double plane given by the branch locus

D4 +

5∑
1

Ti.

From the double cover formulas ([BPV84, Section V. 22]), one obtains K2
S = 8

(notice that the pullback of the lines Ti contains 10 (−1)-curves), χ(OS) = 1 and
pg(S) is equal to the number of generators of the linear system of curves of type
10(6, (2, 3)5T ).

The points p0, . . . , p5 are chosen such that the line T1, through p0, p1, is tangent
to the conic defined by p1, . . . , p5. One can verify that this implies the existence of
the cubic D2. After finding D1, we verify the existence of D2 and D3. The curve
D4 is a general element of the pencil generated by F 1

A, F
2
A, F

3
A.

The difficulty here is the computation of D1. This is done following the steps of
Section 3.2. We omit the details.

We notice that the procedure LinSys can be used to compute the number of
elements of the linear system of curves of type 10(6, (2, 3)5T ), showing that pg(S) = 1,
and to compute the invariant and anti-invariant parts of the bicanonical system
(see Section 2.1), showing the above claim about the bicanonical map of S and
confirming that K2

S = 8.

Appendix A. Magma computations

In this appendix several computations are done using the Computational Algebra
System MAGMA.

The following functions will be useful:

function D(F,i);P:=Parent(F);

return Derivative(F,P.i);end function;

function D2(F,i,j);P:=Parent(F);

return Derivative(Derivative(F,P.i),P.j);end function;

i.e., from now on D(F,i) means ∂F
∂P.i and D2(F,i,j) means ∂2F

∂P.j∂P.i .

A.1. The procedures LinSys and LinSys2.

procedure LinSys(A,d,p,m1,m2,td,~L)

//p,m1,... are tuples of p_i,m1_i,...

x:=A.1;y:=A.2;//The coordinates of A.

L:=LinearSystem(LinearSystem(A,d),p,m1);

for j:=1 to #m2 do

if #Sections(L) eq 0 then break;end if;

a:=p[j][1];b:=p[j][2];

Bup:=[Evaluate(Sections(L)[i],y,(x-a)*y+b) div (x-a)^m1[j]:\

i in [1..#Sections(L)]];//The strict transform of the

//blown-up curves.

L1:=LinearSystem(A,Bup);

L2:=LinearSystem(L1,A![a,td[j]],m2[j]);//Imposing the

//infinitely near singularity.

if #Sections(L2) eq 0 then L:=L2;break;end if;

Bdn:=[Evaluate((x-a)^m1[j]*Sections(L2)[i],y,(y-b)/(x-a)):\

i in [1..#Sections(L2)]];//The blown-down curves.

R:=Universe(Bdn);

//R is a Rational function field. We need a homomorphism

//to send the elements of Bdn into a polynomial ring.

h:=hom<R->CoordinateRing(A)|[x,y]>;

L:=LinearSystem(A,[h(Bdn[i]):i in [1..#Bdn]]);

end for;

end procedure;

procedure LinSys2(A,L,q,m,td,~J)

x:=A.1;y:=A.2;//The coordinates of A.

J:=LinearSystem(L,q,m[1]);

td:=[q[2]] cat td;

for j:=1 to #td-1 do

if #Sections(J) eq 0 then break;end if;

b:=td[j];

Bup:=[Evaluate(Sections(J)[i],y,(x-q[1])*y+b) div\

(x-q[1])^m[j]:i in [1..#Sections(J)]];

J1:=LinearSystem(A,Bup);

J:=LinearSystem(J1,A![q[1],td[j+1]],m[j+1]);

end for;

//

for j:=#td-1 to 1 by -1 do

if #Sections(J) eq 0 then break;end if;

b:=td[j];

Bdn:=[Evaluate((x-q[1])^m[j]*Sections(J)[i],y,(y-b)/\

(x-q[1])):i in [1..#Sections(J)]];

R:=Universe(Bdn);

h:=hom<R->CoordinateRing(A)|[x,y]>;

J:=LinearSystem(A,[h(Bdn[i]):i in [1..#Bdn]]);

end for;

end procedure;

A.2. Double planes.

A.2.1. K2 = 8, 6, 4, 2 and g = 5, 4, 3, 2. Here we have the computations of Section
3.1, with details for the case of a double plane with pg = q = 1 and K2 = 6.

> A<x,y>:=AffineSpace(Rationals(),2);

> p:=[A![7/5,4/5],A![7/5,-4/5],A![2,1],A![2,-1],Origin(A)];

> d:=6;m1:=[2,2,2,2,2];m2:=[2,2,1,1];

> td:=[p[i][2]/p[i][1]:i in [1..#m2]];

> LinSys(A,d,p,m1,m2,td,~L);

> #Sections(L);BaseComponent(L);

5

Scheme over Rational Field defined by

1

We are using symmetry: we want to find p5 = (x, y) and p6 = (x,−y).

> R<x,y,b,c,d,e,n>:=PolynomialRing(Rationals(),7);

> h:=hom<PolynomialRing(L)->R|[x,y]>;

> l:=h(Sections(L));

> F:=l[1]+b*l[2]+c*l[3]+d*l[4]+e*l[5];

> G:=Evaluate(F,y,-y);

> eqF:=2*x*y*D2(F,1,2)+x^2*D2(F,1,1)+y^2*D2(F,2,2);

> eqG:=Evaluate(eqF,y,-y);

The condition eqF = 0 is used to obtain one branch of the double point p5 = (x, y)
tangent to the line T5. In fact, the tangent cone of a plane curve {F (X,Y ) = 0} at
(x, y) is given by

∂2F

∂X2
(x, y)

(X − x)2

2
+

∂2F

∂X∂Y
(x, y)(X − x)(Y − y) +

∂2F

∂Y 2
(x, y)

(Y − y)2

2
.

One can verify that this equation is divisible by xY − yX only if

2xy
∂2F

∂X∂Y
(x, y) + x2 ∂

2F

∂X2
(x, y) + y2

∂2F

∂Y 2
(x, y) = 0.

> dif:=y*(y-1/2*x)*(y-4/7*x)*(y+1/2*x)*(y+4/7*x);

In order to obtain p5, p6 �∈ Ti, i = 1, . . . , 4, and p5 �= p6, we need dif to be different
from zero. This is achieved by imposing the condition 1 + n · dif = 0.

> A:=AffineSpace(R);

> Sch:=Scheme(A,[(x-2)^2+y^2-1,F,D(F,1),D(F,2),G,D(G,1),D(G,2),\

> eqF,eqG,1+n*dif]);

> Dimension(Sch);

0

> PointsOverSplittingField(Sch);

This command gives the points of Sch and the necessary field extensions to de-
fine them. Choosing one of the solutions we obtain points p5, p6 such that there
exists a pencil fA of curves of type 15(7, (4, 4)5T , 4). Let B

′ be a general element of

fA. The branch locus B := B′+
∑5

1 Ti is of type 20(12, (5, 5)
5
T , 4). The correspond-

ing minimal double plane is a surface of general type with pg = q = 1, K2 = 6 and
g = 4.

Verification that B0 := B′ is as stated:

> R<r3>:=PolynomialRing(Rationals());

> K<r3>:=NumberField(r3^4 - 570063504574501/8986626*r3^2+\

> 194676993199491455085153141001/323037787455504);

> x1:=-1225449/218906496039245*r3^2 + 6763320857703/401161\

> 4254780;

> y1:=-7879209182423568/1971150953143623770162761495*r3^3 +

> 37270947258282841632/117281546566527266624785*r3;

> //

> A<x,y>:=AffineSpace(K,2);

> p:=[A![7/5,4/5],A![7/5,-4/5],A![2,1],A![2,-1],A![x1,y1],\

> A![x1,-y1],Origin(A)];

> d:=15;m1:=[4,4,4,4,4,4,7];m2:=[4,4,4,4,4];

> td:=[p[i][2]/p[i][1]:i in [1..#m2]];

> LinSys(A,d,p,m1,m2,td,~L);

> #Sections(L);BaseComponent(L);

2

Scheme over K defined by

1

> B0:=Curve(A,Sections(L)[1]+Sections(L)[2]);

> IsReduced(B0);

true

> Multiplicity(B0,Origin(A));

7

> IsOrdinarySingularity(B0,Origin(A));

true

> [Multiplicity(B0,p[i]):i in [1..6]];

[ 4, 4, 4, 4, 4, 4 ]

> IsOrdinarySingularity(B0,p[6]);

true

> T:=[Curve(A,y-p[i][2]/p[i][1]*x):i in [1..5]];

> [IntersectionNumber(B0,T[i],p[i]):i in [1..5]];

[ 8, 8, 8, 8, 8 ]

> ResolutionGraph(B0,p[1]);

The resolution graph on the Digraph

Vertex Neighbours

1 ([ -2, 4, 1, 0 ]) 2 ;

2 ([ -1, 8, 2, 4 ]) ;

One obtains the same resolution graph for p2, ..., p5.

The curve B0 has been chosen as a general element of L. Since this linear system
has no base component and no additional base points (notice that the resolution
of an element of L at the above base points has self-intersection zero), B0 has no
other singularities.

Now we show that the number of generators of the linear system described in
Proposition 5 is 7, confirming that K2 = 6.

> d:=14;m1:=[2,2,2,2,2,2,10];m2:=[4,4,4,4,4];

> LinSys(A,d,p,m1,m2,td,~L);

> #Sections(L);

7

The other cases, pg = q = 1, K2 = 8, 4, 2 and g = 5, 3, 2, are analogous to the
previous one. One needs only to ask Magma (using the procedure LinSys) for

curves of type

[10 + 2n](2n+ 2, (5, 5)nT , 4
6−n), n = 6, 4, 3,

with singularities at the previous points p0, . . . , p6.

A.2.2. K2 = 7, 5, 3 and g = 5, 4, 3. The detailed computations of Section 3.2 are
as follows.

> A<x,y>:=AffineSpace(Rationals(),2);

> p:=[A![1,0],A![1/5,2/5],A![2/5,1/5],A![8/5,9/5],\

> A![9/5,8/5],Origin(A)];

> d:=15;m1:=[4,4,4,4,4,7];m2:=[4,4,4,4,4];

> td:=[p[i][2]/p[i][1]:i in [1..#m2]];

> LinSys(A,d,p,m1,m2,td,~L);

> #Sections(L);BaseComponent(L);

8

Scheme over Rational Field defined by

1

> Bup:=[Evaluate(Sections(L)[i],y,(x-1)*y+0) div (x-1)^4:\

> i in [1..#Sections(L)]];

> L:=LinearSystem(A,Bup);

> Bup:=[Evaluate(Sections(L)[i],y,(x-1)*y+0) div (x-1)^4:\

> i in [1..#Sections(L)]];

> L1:=LinearSystem(A,Bup);

At this stage we have imposed the (4, 4)-points and resolved p1.

> R<x,y,u,v,n>:=PolynomialRing(Rationals(),5);

> h:=hom<PolynomialRing(L1)->R|[x,y]>;

> l:=h(Sections(L1));

Now we impose, to the elements of l, the necessary conditions in order to obtain
the (3, 3)-point p6 = (u, v). The matrix Mt defined by these conditions cannot have
maximal rank.

> H:=[Evaluate(l[i],[u,v,u,v,n]):i in [1..#l]];

> F:=[Evaluate(l[i],y,(x-u)*y+v):i in [1..#l]];

> G:=[(F[i]-Evaluate(F[i],x,u)) div (x-u):i in [1..#l]];

> G1:=[(G[i]-Evaluate(G[i],x,u)) div (x-u):i in [1..#l]];

> G2:=[(G1[i]-Evaluate(G1[i],x,u)) div (x-u):i in [1..#l]];

> F:=G2;

> M:=[[H[i],D(H[i],3),D(H[i],4),D2(H[i],3,3),D2(H[i],3,4),\

> D2(H[i],4,4),F[i],D(F[i],1),D(F[i],2),D2(F[i],1,1),\

> D2(F[i],1,2),D2(F[i],2,2)]:i in [1..#l]];

> ME:=[[Evaluate(M[i][o],[1,y,1,v,n]):o in [1..12]]:i in [1..#M]];

> //This last step is needed to increase the speed of calculations.

> Mt:=Matrix(ME);

> min:=Minors(Mt,#l);

> A:=AffineSpace(R);

> Sch:=Scheme(A,min cat [x-1,u-1,1+n*v]);

> Dimension(Sch);

0

> PointsOverSplittingField(Sch);

As before, this gives various solutions. We choose one which works. Here are the
verifications:

> R<r2>:=PolynomialRing(Rationals());

> K<r2>:=NumberField(r2^2 - 1292/35);

> A<x,y>:=AffineSpace(K,2);

> p:=[A![1/5,2/5],A![2/5,1/5],A![8/5,9/5],A![9/5,8/5],\

> Origin(A)];

> d:=15;m1:=[4,4,4,4,7];m2:=[4,4,4,4];

> td1:=[p[i][2]/p[i][1]:i in [1..#m2]];

> LinSys(A,d,p,m1,m2,td1,~L);

> //

> q:=A![1,0];m:=[4,4,3,3];

> td2:=[0,35/1292*r2,-455/15504*r2 + 455/7752];

> LinSys2(A,L,q,m,td2,~J);

> #Sections(J);

1

> B0:=Curve(A,Sections(J)[1]);

> IsReduced(B0);

true

> p1:=[q] cat p;

> T:=[Curve(A,y-p1[i][2]/p1[i][1]*x):i in [1..5]];

> [Multiplicity(B0,p1[i]):i in [1..6]];

[ 4, 4, 4, 4, 4, 7 ]

> [IntersectionNumber(T[i],B0,p1[i]):i in [1..5]];

[ 8, 8, 8, 8, 8 ]

> [ResolutionGraph(B0,p1[i]):i in [1,2]];

[

The resolution graph on the Digraph

Vertex Neighbours

1 ([ -2, 8, 2, 1 ]) 2 3 ;

2 ([ -2, 4, 1, 0 ]) ;

3 ([ -2, 11, 3, 0 ]) 4 ;

4 ([ -1, 14, 4, 3 ]) ;

,

The resolution graph on the Digraph

Vertex Neighbours

1 ([ -2, 4, 1, 0 ]) 2 ;

2 ([ -1, 8, 2, 4 ]) ;

]

The resolution graphs for the points p3, p4 and p5 are equal to this last one.

Now we calculate the pencil J which induces the Albanese fibration.

> d:=16;m1:=[4,4,4,4,8];m2:=[4,4,4,4];

> LinSys(A,d,p,m1,m2,td1,~L);

> q:=A![1,0];m:=[4,4,4,4];

> LinSys2(A,L,q,m,td2,~J);

> #Sections(J);BaseComponent(J);

2

Scheme over K defined by

1

> Jy:=LinearSystem(J,Curve(A,y));

> B0 eq Curve(A,Sections(Jy)[1] div y);

true

We show that the curve B0 has no other singularities:

> SingularPoints(B0);

{@ (0, 0), (1/5, 2/5), (2/5, 1/5), (1, 0), (8/5, 9/5), (9/5, 8/5) @}

> HasSingularPointsOverExtension(B0);

false

> PB0:=ProjectiveClosure(B0);

> Dimension(SingularSubscheme(PB0) meet LineAtInfinity(A));

-1

Finally we verify that the number of generators of the linear system described
in Proposition 5 is 8, which implies K2 = 7.

> d:=14;m1:=[2,2,2,2,10];m2:=[4,4,4,4];

> LinSys(A,d,p,m1,m2,td1,~L);

> m:=[2,4,0,2];

> LinSys2(A,L,q,m,td2,~J);

> #Sections(J);

8

With this we have constructed a minimal double plane with pg = q = 1, K2 = 7
and g = 5.

A.2.3. K2 = 6 and g = 3. Here we give the detailed computations of Section 3.3.

> A<x,y>:=AffineSpace(Rationals(),2);

> p:=[A![4/5,7/5],A![-4/5,7/5],A![1,2],A![-1,2],Origin(A)];

> d:=6;m1:=[2,2,2,2,2];m2:=[2,2,2,2];

> td:=[p[i][2]/p[i][1]:i in [1..#m2]];

> LinSys(A,d,p,m1,m2,td,~L);

> #Sections(L);BaseComponent(L);

2

Scheme over Rational Field defined by

1

> R<x,y,b,n>:=PolynomialRing(Rationals(),4);

> h:=hom<PolynomialRing(L)->R|[x,y]>;

> l:=h(Sections(L));

> F:=l[1]+b*l[2];

> G:=Evaluate(F,x,-x);

> C:=x^2+(y-2)^2-1;

> //

> eqF:=D(C,1)*D(F,2)-D(C,2)*D(F,1);//To obtain a curve

> //tangent to the conic C at p_5=(x,y).

> eqG:=Evaluate(eqF,x,-x);//The same to p_6=(-x,y).

> dif:=x*(y-2*x)*(y-7/4*x)*(y+2*x)*(y+7/4*x);

> //We need dif to be non-zero.

> //

> A:=AffineSpace(R);

> Sch:=Scheme(A,[C,F,G,eqF,eqG,1+n*D(F,1)*D(F,2)*dif]);

> Dimension(Sch);

0

> PointsOverSplittingField(Sch);

{@(-r2, 0, 5377/5292, -3/307328), (r2, 0, 5377/5292, -3/307328)@}

Algebraically closed field with 2 variables

Defining relations: [

r2^2 + 3,

r1^2 + 3 ]

This gives the points p5, p6. We omit the remaining verifications (they are similar
to the ones in A.2.1 and A.2.2).

(58:5686)

[Bor07] G. Borrelli, The classification of surfaces of general type with nonbirational bicanonical

map, J. Algebraic Geom. 16 (2007), no. 4, 625–669. MR2357686 (2008m:14074)
[BPV84] W. Barth, C. Peters, and A. Van de Ven, Compact complex surfaces, vol. 4, Springer-

Verlag, Berlin, 1984. MR749574 (86c:32026)
[Cat81] F. Catanese, On a class of surfaces of general type, Algebraic Surfaces, CIME, Liguori,

269–284, 1981.
[Cat99] , Singular bidouble covers and the construction of interesting algebraic surfaces,

Contemp. Math. 241, Amer. Math. Soc., 97–120, 1999. MR1718139 (2000j:14061)
[CC91] F. Catanese and C. Ciliberto, Surfaces with pg = q = 1, Sympos. Math. 32, Academic

Press, 49–79, 1991. MR1273372 (95d:14030)
[CC93] , Symmetric products of elliptic curves and surfaces of general type with pg =

q = 1, J. Algebraic Geom. 2 (1993), no. 3, 389–411. MR1211993 (94i:14040)
[Cil97] C. Ciliberto, The bicanonical map for surfaces of general type, Proc. Sympos. Pure

Math. 62.1, Kollár, János et al. (eds.), Algebraic geometry, 57–84, 1997. MR1492518
(98m:14040)

[CM02] C. Ciliberto and M. Mendes Lopes, On surfaces with pg = q = 2 and non-birational
bicanonical map, Adv. Geom. 2 (2002), no. 3, 281–300. MR1924760 (2004d:14053)

[CP06] F. Catanese and R. Pignatelli, Fibrations of low genus. I, Ann. Scient. École. Norm.

